Décompositions LU et Choleski

(Programmation avec Maple)

Préparation à la nouvelle épreuve d'informatique
de l'École Polytechnique.

Jean-Michel Ferrard

Énoncé

Ce document vise à la préparation à la nouvelle épreuve d'informatique de l'École Polytechnique.

Les caractéristiques de cette épreuve peuvent être consultées à l'adresse :

http ://www.enseignement.polytechnique.fr/informatique/concours/

Parmi les langages de programmation possibles, on a choisi Maple, dont on n'a utilisé que les fonctionnalités de base, conformément aux demandes des concepteurs de l'épreuve.

I. La décomposition LU

Soit A une matrice inversible de $\mathcal{M}_n(\mathbb{K})$. Une *décomposition LU* de A est une égalité $A = LU$, où L est une matrice triangulaire inférieure (L pour "Low") à diagonale unité (tous les coefficients diagonaux valent 1), et où U est une matrice triangulaire supérieure (U pour "Up"). La matrice U est nécessairement inversible (donc ses coefficients diagonaux sont non nuls.)

Par exemple, $A = \begin{pmatrix} 2 & -3 & 1 & -1 \\ -2 & 2 & -3 & 2 \\ 4 & -9 & -2 & 3 \\ -2 & 5 & 5 & -4 \end{pmatrix} = \begin{pmatrix} 1 & 0 & 0 & 0 \\ -1 & 1 & 0 & 0 \\ 2 & 3 & 1 & 0 \\ -1 & -2 & 1 & 1 \end{pmatrix} \begin{pmatrix} 2 & -3 & 1 & -1 \\ 0 & -1 & -2 & 1 \\ 0 & 0 & 2 & 2 \\ 0 & 0 & 0 & -5 \end{pmatrix}$

Pour tout indice k compris entre 1 et n, on notera A_k la matrice extraite de A formée à l'intersection des k premières lignes et des k premières colonnes de A. On notera Δ_k le déterminant de A_k. Les Δ_k sont appelés les *mineurs principaux* de A.

Proposition *(Existence et unicité de la décomposition LU)*

> Une matrice A de $\mathcal{GL}_n(\mathbb{K})$ possède une décomposition LU si et seulement si ses mineurs principaux sont non nuls. Cette décomposition est alors unique. [Démonstration]

Dans la suite de cette partie, on suppose que la matrice A possède une décomposition LU et on voit comment mettre en œuvre le calcul des matrices L et U.

On note a_{ij}, ℓ_{ij}, et u_{ij} les termes généraux de A, L, U.

1. Ecrire les égalités donnant a_{ik} en fonction des ℓ_{ij} (avec $j \leq i$) et u_{jk} (avec $j \leq k$).

   En déduire les expressions :

   – De u_{ij}, pour $i \leq j$, en fonction de a_{ij}, de ℓ_{ik} $(k < i)$, de u_{kj} $(k < i)$.

   – De ℓ_{ij}, pour $i > j$, en fonction de a_{ij}, de ℓ_{ik} $(k < j)$, de u_{kj} $(k \leq j)$.

   Montrer comment les égalités obtenues permettent de calculer de proche en proche (et on précisera dans quel ordre) tous les coefficients de L et de U. [S]

2. En déduire une procédure `LU1` calculant la décomposition d'une matrice A (supposée carrée inversible d'ordre n, et possédant une telle décomposition).

   La syntaxe sera `LU1`$(A, `L`, `U`, n)$ et les deux matrices L et U seront placées dans les noms de variable homonymes passés en argument. [S]

3. Dans un souci d'économie, on décide de former une seule matrice B, carrée d'ordre n, dont le terme général b_{ij} est donné par : $b_{ij} = u_{ij}$ si $i \le j$, et $b_{ij} = \ell_{ij}$ si $i > j$.

   (a) Montrer que B peut être formée de la manière suivante, dans l'ordre des j croissants :
   $\forall i \in \{1, \ldots, n\}, \quad b_{ij} = a_{ij} - \sum_{k=1}^{\min(i,j)-1} b_{ik}\, b_{kj}$ suivi de $b_{ij} \leftarrow \dfrac{b_{ij}}{b_{jj}}$ si $i > j$.
   Montrer qu'il est possible, dans le programme, de n'utiliser qu'une variable pour désigner la matrice initiale A, puis pour former petit à petit la matrice B. [S]

   (b) En déduire une procédure `LU2` décomposant la matrice A, et plaçant le résultat dans la matrice A elle-même (les coefficients sur et au-dessus de la diagonale étant ceux de U et les coefficients sous la diagonale étant ceux de L.) [S]

4. *Décomposition $PA = LU$.*

   On a vu que l'existence de la décomposition $A = LU$ est soumise à des conditions sur la matrice inversible A (la première d'entre elles étant que a_{11} doit être non nul.)

   Chacune de ces conditions équivaut à la non-nullité d'un coefficient utilisé comme pivot. Si un tel pivot est nul, la décomposition est impossible, à moins qu'on ne puisse continuer moyennant un échange de lignes.

   Si ce pivot est "presque nul", on utilisera encore un échange de lignes (sinon la division par le pivot conduit à une imprécision importante.)

   Ces considérations conduisent à envisager la décomposition LU non pas de la matrice A initiale, mais d'une matrice $B = PA$, où P est une *matrice de permutation*.

   Soit P un élément de $\mathcal{M}_n(\mathbb{K})$, dont le coefficient d'indice (i, j) est noté p_{ij}.

   On dit que P est une matrice de permutation s'il existe une permutation σ de $\{1, 2, \ldots, n\}$ telle que $p_{ij} = \delta_{\sigma(i)j}$ (notations de Kronecker), pour tous indices i et j.

   Pour toute matrice A de $\mathcal{M}_n(\mathbb{K})$, la matrice $B = PA$ se déduit de A par la permutation σ appliquée aux lignes de A : pour tout i de $\{1, \ldots, n\}$, la ligne d'indice i de $B = PA$ est en effet la ligne d'indice $\sigma(i)$ de A.

   Proposition *(Existence de la décomposition $PA = LU$)*

   Pour tout matrice A de $\mathcal{GL}_n(\mathbb{K})$, il existe au moins une matrice de permutation P telle que la matrice PA admette une décomposition LU. [Démonstration]

   On a vu dans la question précédente que la décomposition LU de A pouvait s'effectuer directement sur cette matrice. Cette décomposition s'effectue en n étapes successives (une par colonne.)

   Pour tout j de $\{1, \ldots, n\}$, l'étape j se déroule en deux phases :

   – Pour tout i de $\{1, \ldots, n\}$ on réalise l'affectation $a_{ij} \leftarrow a_{ij} - \sum_{k=1}^{\min(i,j)-1} a_{ik}a_{kj}$.

     On modifie ainsi tous les coefficients de la j-ème colonne.

     Le pivot est maintenant le nouveau coefficient diagonal $p = a_{jj}$.

– Pour tout i de $\{j+1,\ldots,n\}$ (donc sous la diagonale) on divise a_{ij} par le pivot p.

Plaçons nous après la première phase de l'étape j. Notons A' l'état actuel de la matrice dans laquelle s'effectue la décomposition, et A l'état initial de cette matrice.
Donnons-nous un indice de ligne $i \geq j$.

Notons $\mathrm{L}_1,\ldots,\mathrm{L}_n$ les lignes de A et $\mathrm{L}'_1,\ldots,\mathrm{L}'_n$ celles de A'.

Il n'est pas difficile de se convaincre (en remontant les calculs effectués pour en arriver à cette phase de la décomposition) que la ligne L'_i ne dépend que des lignes $\mathrm{L}_1,\ldots,\mathrm{L}_{j-1},\mathrm{L}_i$. Notons par exemple $\mathrm{L}'_i = \varphi(\mathrm{L}_1,\ldots,\mathrm{L}_{j-1},\mathrm{L}_i)$.

On se rend compte également que cette fonction φ est indépendante de l'indice $i \geq j$.

Cela est valable en particulier pour l'indice j lui-même.

On en déduit que si on veut que soient échangées les lignes L'_j et L'_i (à ce stade de la décomposition) il suffit d'échanger les lignes L_j et L_i dans la matrice initiale. Il est donc possible de procéder, à l'issue de la première phase de l'étape j, à l'échange de L'_j avec une ligne L'_i (où $i > j$) de façon à rendre maximum (en valeur absolue) le pivot p qui sera utilisé dans la deuxième phase de cette étape.

Il suffit alors de mémoriser les échanges de lignes effectués tout au long de la décomposition. On aura ainsi obtenu la décomposition LU de la matrice PA, où P est la matrice de permutation résumant les échanges successifs de lignes.

Si cette méthode échoue, c'est-à-dire si le pivot maximum obtenu à un moment donné est nul, c'est que A n'a pas de décomposition $PA = LU$: c'est qu'elle n'est pas inversible.

(a) Écrire une procédure `LU` décomposant la matrice A, et plaçant le résultat dans la matrice A elle-même (les coefficients sur et au-dessus de la diagonale étant ceux de U et les coefficients sous la diagonale étant ceux de L.)

La procédure `LU` se distinguera de `LU2` par la recherche d'un pivot de valeur absolue maximum sur chaque colonne (suivie éventuellement d'un échange de lignes.)

On utilisera la syntaxe `LU`$(A, \text{‘}s\text{‘}, n)$: dans cette écriture la variable s recevra la permutation représentant la succession des échanges de lignes, sous la forme d'un tableau unidimensionnel $[s(1),\ldots,s(n)]$. [S]

(b) Montrer comment ce qui précède permet de ramener la résolution d'un système linéaire $AX = B$ (où A est carrée d'ordre n et inversible, et où B est un vecteur colonne de hauteur n) à la résolution de deux systèmes triangulaires successifs.

Écrire une fonction `syst` renvoyant la solution X du système $AX = B$, sous la forme d'un vecteur de taille n. La syntaxe sera `syst`(M, B, s, n), où :

– M est la matrice carrée contenant la décomposition LU.

– B est un vecteur représentant les seconds membres.

– s est un vecteur représentant la permutation P telle que $PA = LU$. [S]

(c) Vérifier sur un exemple numérique de résolution de système (en imposant par exemple une valeur très faible pour a_{11} dans A) que la décomposition obtenue par `LU` peut se révéler beaucoup plus précise que celle qui est obtenue par `LU2`. [S]

(d) Écrire une fonction `inv` inversant une matrice carrée A d'ordre n. [S]

II. La décomposition de Choleski

Les matrices considérées ici sont à coefficients réels. On identifiera un vecteur u de $\mathbb{R}^n$ avec la matrice uni-colonne U de ses coordonnées dans la base canonique (e).

Le produit scalaire canonique de $\mathbb{R}$ peut donc s'écrire : $< u, v >= {}^tUV = \sum_{k=1}^{n} u_k\, v_k$.

Soit S une matrice carrée d'ordre n, symétrique et à coefficients réels. On sait que S est diagonalisable sur $\mathbb{R}$, et que ses différents sous-espaces propres sont orthogonaux deux à deux.

En particulier, il existe une base orthonormée $(\varepsilon) = \varepsilon_1, \ldots, \varepsilon_n$ formée de vecteurs propres de S.

Proposition (matrices symétriques définies positives)

> Soit S une matrice carrée d'ordre n, symétrique et à coefficients réels.
>
> Les conditions suivantes sont équivalentes :
>
> – Toutes les valeurs propres de S sont strictement positives.
>
> – Pour tout vecteur non nul X de $\mathbb{R}^n$, ${}^tXSX > 0$.
>
> – Il existe une matrice inversible M telle que $S = M\,{}^tM$.
>
> Si ces conditions sont réalisées, on dit S est *définie positive*. [Démonstration]

La matrice inversible M telle que $S = M\,{}^tM$ n'est pas unique, à moins de lui en demander "un peu plus". C'est l'objet de la proposition suivante, qui introduit la décomposition de Choleski.

Proposition *(décomposition de Choleski)*

> Soit S une matrice symétrique définie positive. Il existe une unique matrice L triangulaire inférieure à coefficients diagonaux strictement positifs, telle que $S = L\,{}^tL$.
>
> Cette écriture est appelée *décomposition de Choleski* de S. [Démonstration]

1. Soit $S = L\,{}^tL$ la décomposition de Choleski d'une matrice S définie positive.
   On note s_{ij} et ℓ_{ij} les coefficients d'indice i, j de S et L.
   Établir un système d'égalités permettant de calculer les coefficients ℓ_{ij}. [S]

2. Écrire une fonction `choleski` prenant en argument une matrice carrée M à coefficients réels, supposée définie positive, et renvoyant la matrice L de la décomposition de M. [S]

Corrigé

I. La décomposition LU

1. Le terme général a_{ij} de la matrice $A = LU$ s'écrit $a_{ij} = \sum_{k=1}^{n} \ell_{ik}\, u_{kj}$.

   Or $\ell_{ik} = 0$ quand $k > i$ et $u_{kj} = 0$ quand $k > j$.

   La somme est donc limitée à $k = \min(i,j)$: $a_{ij} = \sum_{k=1}^{\min(i,j)} \ell_{ik}\, u_{kj}$. On en déduit :

   Si $i \leq j$, alors $a_{ij} = \sum_{k=1}^{i} \ell_{ik}\, u_{kj} = \sum_{k=1}^{i-1} \ell_{ik}\, u_{kj} + u_{ij} \Rightarrow u_{ij} = a_{ij} - \sum_{k=1}^{i-1} \ell_{ik}\, u_{kj} \quad (1)$

   Si $i > j$: $a_{ij} = \sum_{k=1}^{j} \ell_{ik}\, u_{kj} = \sum_{k=1}^{j-1} \ell_{ik}\, u_{kj} + \ell_{ij}\, u_{jj} \Rightarrow \ell_{ij} = \dfrac{1}{u_{jj}}\Big(a_{ij} - \sum_{k=1}^{j-1} \ell_{ik}\, u_{kj}\Big) \quad (2)$

   Ces égalités permettent de former L et U, ligne par ligne, ou colonne par colonne.

   – Pour $j = 1$, l'égalité (1) donne $u_{11} = a_{11}$ (non nul car $a_{11} = \Delta_1$.)
     Puis (2) donne la *sous-diagonale* de la colonne 1 de L : $\forall\, i \in \{2, \ldots, n\}$, $\ell_{i1} = \dfrac{a_{i1}}{u_{11}}$.

   – Soit $j \geq 2$. Supposons connues les colonnes 1 à $j-1$ de L et U.
     Quand i varie de 1 à j, l'égalité (1) donne la j-ème colonne de U.
     Quand i varie de $j+1$ à n, l'égalité (2) donne la j-ème colonne de L.
     La division par u_{jj} est possible car par hypothèse la décomposition $A = LU$ existe.
     En tout cas, tout nouveau coefficient est obtenu en fonction de coefficients déjà calculés !

   On peut donc calculer de proche en proche tous les coefficients de U et L, en procédant de la colonne 1 à la colonne n, et sur chaque colonne de la ligne 1 à la ligne n. [Q]

2. La procédure LU1 est la traduction des calculs précédents :

```
> LU1:=proc(A::array,L::name,U::name,n::integer)
>   local i,j,k,c:
>   L:=matrix(n,n,0); U:=matrix(n,n,0);  # matrices nulles d'ordre n
>   for j to n do                  # boucle de formation des matrices L et U
>     for i to j do                # calcul de la colonne j de U
>       c:=A[i,j];                 # initialise le coefficient d'indice i,j
>       for k to i-1 do            # boucle de calcul de U[i,j]
>         c:=c-L[i,k]*U[k,j];
>       od;
>       U[i,j]:=c;                 # place le coefficient dans U
>     od;
>     L[j,j]:=1;                   # coefficient diagonal d'indice j de L
>     for i from j+1 to n do       # calcul de la colonne j de L
>       c:=A[i,j];                 # initialise le coefficient d'indice i,j
>       for k to j-1 do            # boucle de calcul de L[i,j]
>         c:=c-L[i,k]*U[k,j]
```

```
>       od;
>       L[i,j]:=c/U[j,j];          # place le coefficient dans L
>     od;
>   od;
> end:
```

On teste LU1 sur la matrice A citée comme exemple dans l'énoncé. On affiche ensuite L et U, et on vérifie l'égalité $A = LU$:

```
> A:=matrix([[2,-3,1,-1],[-2,2,-3,2],[4,-9,-2,3],[-2,5,5,-4]]):
> LU1(A,'L','U',4): eval(L), eval(U), evalm(L&*U);
```

[Q] $$\begin{pmatrix} 1 & 0 & 0 & 0 \\ -1 & 1 & 0 & 0 \\ 2 & 3 & 1 & 0 \\ -1 & -2 & 1 & 1 \end{pmatrix}, \begin{pmatrix} 2 & -3 & 1 & -1 \\ 0 & -1 & -2 & 1 \\ 0 & 0 & 2 & 2 \\ 0 & 0 & 0 & -5 \end{pmatrix}, \begin{pmatrix} 2 & -3 & 1 & -1 \\ -2 & 2 & -3 & 2 \\ 4 & -9 & -2 & 3 \\ -2 & 5 & 5 & -4 \end{pmatrix}$$

3. Décomposition “in situ”.

   (a) Dans les égalités (1) et (2) de la première question, tous les coefficients de L et U peuvent être renommés à l'aide de la convention suivante :

   Pour tous indices i, j notons $b_{ij} = u_{ij}$ si $i \leq j$ et $b_{ij} = \ell_{ij}$ si $i > j$.

   On en déduit que les égalités (1) et (2) peuvent être réécrites sous la forme :

   (1) Si $i \leq j$, alors $b_{ij} = a_{ij} - \sum_{k=1}^{i-1} b_{ik}\, b_{kj}$ (2) Si $i > j$, alors $b_{ij} = \frac{1}{b_{jj}}\Big(a_{ij} - \sum_{k=1}^{j-1} b_{ik}\, b_{kj}\Big)$

   Il revient au même d'écrire, dans l'ordre croissant des indices de colonne j :

   Pour tout i de $\{1, \ldots, n\}$, $b_{ij} = a_{ij} - \sum_{k=1}^{\min(i,j)-1} b_{ik}\, b_{kj}$, suivi de $b_{ij} \leftarrow \dfrac{b_{ij}}{b_{ii}}$ si $i > j$.

   Chaque a_{ij} n'est utilisé que lors du calcul de b_{ij}. On peut donc remplacer au fur et à mesure les a_{ij} par les b_{ij} et ainsi n'utiliser que la variable A. [Q]

   (b) Voici la procédure LU2, qui se déduit très simplement des considérations précédentes :

```
> LU2:=proc(A::array,n)
>   local i,j,k,p;
>   for j to n do              # calcul de la colonne j
>     for i to n do             # calcul du terme d'indice i,j
>       for k to min(i,j)-1 do
>         A[i,j]:=A[i,j]-A[i,k]*A[k,j]
>       od;
>     od;
>     p:= A[j,j];               # c'est le pivot de la colonne j
>     for i from j+1 to n do    # modification des termes sous-diagonaux
>       A[i,j]:=A[i,j]/p
>     fi;
>   od;
> end:
```

On décompose la même matrice. Le résultat, dans A, est une superposition de L, U comme on le voit en rappelant ces matrices obtenues avec `LU1` :

```
> LU2(A,4): eval(A), eval(L), eval(U);
```

$$\begin{pmatrix} 2 & -3 & 1 & -1 \\ -1 & -1 & -2 & 1 \\ 2 & 3 & 2 & 2 \\ -1 & -2 & 1 & -5 \end{pmatrix}, \begin{pmatrix} 1 & 0 & 0 & 0 \\ -1 & 1 & 0 & 0 \\ 2 & 3 & 1 & 0 \\ -1 & -2 & 1 & 1 \end{pmatrix}, \begin{pmatrix} 2 & -3 & 1 & -1 \\ 0 & -1 & -2 & 1 \\ 0 & 0 & 2 & 2 \\ 0 & 0 & 0 & -5 \end{pmatrix}$$

[Q]

4. *Décomposition $PA = LU$.*

   (a) Voici la procédure `LU` :

```
> LU:=proc(A::array,s::name,n::integer)
>   local i,j,k,p,q,t;
>   s:=array(1..n);
>   for i to n do s[i]:=i od;      # par défaut, s = permutation identité
>   for j to n do                  # de la colonne j = 1 à la colonne j = n
>     for i to n do                  # de haut en bas sur la colonne j
>       for k to min(i,j)-1 do       # boucle de calcul de A[i,j]
>         A[i,j]:=A[i,j]-A[i,k]*A[k,j]
>       od;
>     od;
>     p:=A[j,j]; i:=j;               # valeur et ligne initiales du pivot
>     for k from j+1 to n do         # boucle de recherche du meilleur pivot
>       if abs(A[k,j])>abs(p) then p:=A[k,j]; i:=k; fi;
>     od;
>     if p=0 then                    # si pivot=0, matrice non inversible
>        ERROR("matrice non inversible")
>     fi;
>     if i>j then                    # si le pivot n'est pas diagonal
>       for k to n do                # boucle d'échange des lignes j et i
>         t:=A[j,k]: A[j,k]:=A[i,k]: A[i,k]:=t
>       od;
>       t:=s[j]; s[j]:=s[i]; s[i]:=t;  # mémorise l'échange des lignes
>     fi;
>     for k from j+1 to n do A[k,j]:=A[k,j]/p od;   # divise par le pivot
>   od;
> end:
```

On reprend la matrice utilisée pour illustrer le fonctionnement de `LU1` et `LU2`.

On la décompose avec `LU`, et on affiche la permutation des lignes, et le résultat de la décomposition (L et U sont superposées dans la variable initiale A.)

```
> A:=matrix([[2,-3,1,-1],[-2,2,-3,2],[4,-9,-2,3],[-2,5,5,-4]]):
> LU(A,'s',4); eval(s),eval(A);
```

$$[3,2,4,1], \quad \begin{pmatrix} 4 & -9 & -2 & 3 \\ -1/2 & -5/2 & -4 & 7/2 \\ -1/2 & -1/5 & 16/5 & -9/5 \\ 1/2 & -3/5 & -1/8 & -5/8 \end{pmatrix}$$

Ce résultat montre que ce n'est pas A qui a été décomposée, mais $B = PA$, avec :

$$B = PA = \begin{pmatrix} 0 & 0 & 1 & 0 \\ 0 & 1 & 0 & 0 \\ 0 & 0 & 0 & 1 \\ 1 & 0 & 0 & 0 \end{pmatrix} \begin{pmatrix} 2 & -3 & 1 & -1 \\ -2 & 2 & -3 & 2 \\ 4 & -9 & -2 & 3 \\ -2 & 5 & 5 & -4 \end{pmatrix} = \begin{pmatrix} 4 & -9 & -2 & 3 \\ -2 & 2 & -3 & 2 \\ -2 & 5 & 5 & -4 \\ 2 & -3 & 1 & -1 \end{pmatrix}$$

Cela est confirmé en appliquant la procédure `LU1` à la matrice B :

```
> B:=matrix([[4,-9,-2,3],[-2,2,-3,2],[-2,5,5,-4],[2,-3,1,-1]]):
> LU1(B,'L','U',4); eval(L), eval(U);
```

$$\begin{pmatrix} 1 & 0 & 0 & 0 \\ -1/2 & 1 & 0 & 0 \\ -1/2 & -1/5 & 1 & 0 \\ 1/2 & -3/5 & -1/8 & 1 \end{pmatrix}, \quad \begin{pmatrix} 4 & -9 & -2 & 3 \\ 0 & -5/2 & -4 & 7/2 \\ 0 & 0 & 16/5 & -9/5 \\ 0 & 0 & 0 & -5/8 \end{pmatrix}$$

[Q]

(b) Il existe une matrice de permutation P, et deux matrices L, U telles que $PA = LU$.
Le système $AX = B$ est alors équivalent à $PAX = PB$, c'est-à-dire à $LUX = PB$.
Si P est associée à la permutation σ, alors $B' = PB = (b_{\sigma(1)}, b_{\sigma(2)}, \ldots, b_{\sigma(n)})$.
Dans ces conditions, trouver X c'est résoudre $LY = B'$ puis $UX = Y$.
Ainsi le système initial se ramène à deux systèmes triangulaires successifs.
Posons $X = (x_1, x_2 \ldots, x_n)$ et $Y = (y_1, y_2, \ldots, y_n)$. Le système $LY = B'$ s'écrit :

$$\begin{pmatrix} 1 & 0 & 0 & \ldots & 0 \\ \ell_{21} & 1 & 0 & \ldots & 0 \\ \ell_{31} & \ell_{32} & 1 & \ddots & 0 \\ \vdots & \vdots & \ddots & \ddots & 0 \\ \ell_{n1} & \ell_{n2} & \ldots & \ell_{n,n-1} & 1 \end{pmatrix} \begin{pmatrix} y_1 \\ y_2 \\ y_3 \\ \vdots \\ y_n \end{pmatrix} = \begin{pmatrix} b_{\sigma(1)} \\ b_{\sigma(2)} \\ b_{\sigma(3)} \\ \vdots \\ b_{\sigma(n)} \end{pmatrix} \Leftrightarrow \begin{cases} y_1 = b_{\sigma(1)} \\ y_2 = b_{\sigma(2)} - \ell_{21} y_1 \\ y_3 = b_{\sigma(3)} - \ell_{31} y_1 - \ell_{32} y_2 \\ \ldots = \ldots \\ y_n = b_{\sigma(n)} - \sum_{j=1}^{n-1} \ell_{nj} y_j \end{cases}$$

Ensuite le système $UX = Y$ s'écrit :

$$\begin{pmatrix} u_{11} & u_{12} & u_{13} & \ldots & u_{1n} \\ 0 & u_{22} & u_{23} & \ldots & u_{2n} \\ 0 & 0 & u_{33} & \ldots & u_{3n} \\ \vdots & \vdots & \ddots & \ddots & \vdots \\ 0 & 0 & \ldots & 0 & u_{nn} \end{pmatrix} \begin{pmatrix} x_1 \\ x_2 \\ x_3 \\ \vdots \\ x_n \end{pmatrix} = \begin{pmatrix} y_1 \\ y_2 \\ y_3 \\ \vdots \\ y_n \end{pmatrix} \Leftrightarrow \begin{cases} x_n = y_n / u_{nn} \\ x_{n-1} = \dfrac{y_{n-1} - u_{n-1,n} y_n}{u_{n-1,n-1}} \\ \ldots = \ldots \\ x_1 = \dfrac{1}{u_{11}} \Big(y_1 - \sum_{j=2}^{n} u_{1j} \, x_j \Big) \end{cases}$$

Voici le listing de `syst`. La variable X désigne successivement les seconds membres (après permutation), la solution Y du premier système, puis la solution finale.

```
> syst:=proc(A::array,B::array,s::array,n::integer)
>   local X,i,j,t;
>   X:=array(1..n);                       # initialisation
>   for i to n do X[i]:=B[s[i]] od;       # permutation sur le second membre
>   for i from 2 to n do                  # résolution du système LY = B
>     t:=0; for j to i-1 do t:=t+A[i,j]*X[j] od; X[i]:=X[i]-t;
>   od;
>   for i from n to 1 by -1 do            # résolution du système UX = Y
>     t:=0; for j from i+1 to n do t:=t+A[i,j]*X[j] od;
>     X[i]:=(X[i]-t)/A[i,i];
>   od; eval(X);                          # renvoie la solution du système
> end:
```

On reprend ici une matrice plusieurs fois utilisée avec les seconds membres $(1, 6, 5, 2)$.

On constate que la solution du système est $X = (-17, -10, -2, -7)$.

```
> A:=matrix([[2,-3,1,-1],[-2,2,-3,2],[4,-9,-2,3],[-2,5,5,-4]]):
> AA:=copy(A): B:=array([1,6,5,2]):
> LU(A,'s',4); X:=syst(A,B,s,4);
```

$$X := [-17, -10, -2, -7]$$

On vérifie maintenant que la solution est correcte. Le produit AX redonne bien B.

```
> evalm(AA&*X);
```

$$[1, 6, 5, 2]$$

[Q]

(c) On reprend l'exemple précédent, mais on remplace $A[1, 1] = 2$ par $\dfrac{1}{1000000}$.

```
> A:=matrix([[2,-3,1,-1],[-2,2,-3,2],[4,-9,-2,3],[-2,5,5,-4]]):
> A[1,1]:=1/1000000: A1:=copy(A): A2:=copy(A):
> B:=array([1,6,5,2]):
```

On résout le système $AX = B$ après `LU2` et `LU` de façon exacte.

On constate (c'est normal) que les résultats sont les mêmes.

On calcule aussi une valeur approchée de la solution exacte (12 chiffres significatifs.)

```
> LU(A1,'s',4); X:=syst(A1,B,s,4);
> LU2(A2,4); X:=syst(A2,B,array([1,2,3,4]),4);
> map(evalf,X);
```

$$X := \left[-\frac{340000000}{61999979}, -\frac{-76000062}{61999979}, -\frac{487999736}{61999979}, -\frac{653999603}{61999979}\right]$$

$$X := \left[-\frac{340000000}{61999979}, -\frac{-76000062}{61999979}, -\frac{487999736}{61999979}, -\frac{653999603}{61999979}\right]$$

$$[-5.483872825, -1.225807867, 7.870966150, 10.54838427]$$

On met maintenant tous les coefficients de A au format “virgule flottante”.

On calcule la solution du système après `LU` puis après `LU2`.

On constate des erreurs d'arrondi assez sensibles dans le premier cas, alors que la fonction `LU` donne un résultat extrêmement précis. On remarque que dans ce dernier cas, on a dû préciser que la permutation s est égale à l'identité.

```
> A:=map(evalf,A): A1:=copy(A): A2:=copy(A):
> LU(A1,'s',4); X:=syst(A1,B,s,4);
> LU2(A2,4); X:=syst(A2,B,array([1,2,3,4]),4);
```

$$X := [-5.483872825, -1.225807867, 7.870966153, 10.54838427]$$

$$X := [-5.490000000, -1.225966469, 7.872628719, 10.55052264]$$

[Q]

(d) Voici la fonction `inv`. L'idée est évidemment de résoudre le système $AX = B$, où les second membres B sont successivement l'une des n colonnes de la matrice identité.

```
> inv:=proc(A::array,n::integer)
>   local B,M,AA,s,i,j,t;
>   M:=array(1..n,1..n); AA:=copy(A); # initialisations
>   LU(AA,'s',n);                     # décomposition PA = LU
>   for j to n do                     # de la colonne 1 à la colonne n
>     B:=vector(n,0); B[j]:=1;          # j-ième colonne de l'identité
>     B:=syst(AA,B,s,n);                # résolution du système
>     for i to n do M[i,j]:=B[i] od;    # j-ième colonne de l'inverse de A
>   od;
>   RETURN(eval(M));                  # renvoie la matrice inverse de M
> end:
```

On reprend la matrice A ayant servi d'exemple jusqu'ici, et on calcule son inverse avec notre fonction `inv`, puis avec la fonction intégrée de Maple. On constate que les résultats sont identiques.

```
> A:=matrix([[2,-3,1,-1],[-2,2,-3,2],[4,-9,-2,3],[-2,5,5,-4]]):
> inv(A,4), evalm(A&^(-1));
```

$$\begin{pmatrix} -\frac{21}{20} & -\frac{7}{4} & -\frac{13}{20} & -\frac{11}{10} \\ -\frac{4}{5} & -1 & -\frac{2}{5} & -\frac{3}{5} \\ -\frac{9}{10} & -\frac{1}{2} & \frac{3}{10} & \frac{1}{5} \\ -\frac{8}{5} & -1 & \frac{1}{5} & -\frac{1}{5} \end{pmatrix}, \begin{pmatrix} -\frac{21}{20} & -\frac{7}{4} & -\frac{13}{20} & -\frac{11}{10} \\ -\frac{4}{5} & -1 & -\frac{2}{5} & -\frac{3}{5} \\ -\frac{9}{10} & -\frac{1}{2} & \frac{3}{10} & \frac{1}{5} \\ -\frac{8}{5} & -1 & \frac{1}{5} & -\frac{1}{5} \end{pmatrix}$$

[Q]

II. La décomposition de Choleski

1. L'égalité $L\,{}^tL = S$ permet par identification de calculer les $\ell_{i,j}$:

$$S = L\,{}^tL \Leftrightarrow \forall i,j \in \{1,\ldots,n\},\ s_{i,j} = \sum_{k=1}^{n} [L]_{i,k}[{}^tL]_{k,j} = \sum_{k=1}^{n} \ell_{i,k}\ell_{j,k} = \sum_{k=1}^{\min(i,j)} \ell_{i,k}\ell_{j,k}$$

La somme est limitée supérieurement à $\min(i,j)$ car L est triangulaire inférieure.

D'autre part, on sait que la matrice S est symétrique. L'identification dans $S = L\,{}^tL$ peut donc se limiter, sans perdre aucune généralitén, au cas $i \geq j$ (coefficients diagonaux et sub-diagonaux). Pour tout entier j fixé (compris entre 1 et n) on trouve :

– (1) : Pour $i = j$: $\displaystyle\sum_{k=1}^{j} \ell_{j,k}^{\,2} = s_{j,j} \Rightarrow \ell_{j,j}^{\,2} = s_{j,j} - \sum_{k=1}^{j-1} \ell_{j,k}^{\,2}$

– (2) : Pour $j+1 \leq i \leq n$: $\displaystyle\sum_{k=1}^{j} \ell_{i,k}\ell_{j,k} = s_{i,j} \Rightarrow \ell_{i,j} = \frac{1}{\ell_{j,j}}\Big(s_{i,j} - \sum_{k=1}^{j-1}\ell_{i,k}\ell_{j,k}\Big)$.

Les coefficients de L peuvent donc être calculés colonne par colonne.

Plus précisément, lors du calcul de la j-ème colonne de L :

– La phase (1) donne le j-ème coefficient diagonal $\ell_{j,j} > 0$ (si le second membre n'est pas strictement positif, c'est que S n'est pas définie positive). On constate que $\ell_{j,j}$ est obtenu en fonction de $s_{j,j}$ et de coefficients déja calculés dans L (sur la même ligne, mais sur les colonnes précédentes).

– La phase (2) donne les termes subdiagonaux de la j-ème colonne de L. Chaque $\ell_{i,j}$ est obtenu en fonction de $s_{i,j}$ (en même position dans S), du coefficient diagonal $\ell_{j,j}$ (qui vient d'être calculé) et de termes déjà connus dans L (sur les lignes i, j, mais sur des colonnes d'indice inférieur à j).

[Q]

2. Voici la fonction `choleski`

```
> choleski:=proc(S::array,n::integer)
>   local L,i,j,k,t:
>   L:=matrix(n,n,0);         # matrice nulle d'ordre n
>   for j to n do             # calcul de L, colonne par colonne
>     t:=S[j,j];                # j-ème coefficient diagonal de S
>     for k to j-1 do           # calcule le j-ème coefficient diagonal de L
>       t:=t-L[j,k]^2;
>     od;
>     if t<=0 then              # erreur si S n'est pas définie positive
>       ERROR("matrice non def+")
>     fi;
>     t:=sqrt(t); L[j,j]:=t;    # place le j-ème coefficient diagonal de L
>     for i from j+1 to n do    # coefficients sous-diagonaux, colonne j
>       t:=S[i,j];
>       for k to j-1 do
```

```
>         t:=t-L[i,k]*L[j,k];
>       od;
>       L[i,j]:=t/L[j,j];       # place le coefficient d'indice (i,j) dans L
>     od;
>   od;
>   RETURN(eval(L));           # renvoie la matrice L
> end:
```

On crée une matrice symétrique S :

```
S:=matrix([[1,-2,4],[-2,13,-11],[4,-11,21]]);
```

$$S := \begin{bmatrix} 1 & -2 & 4 \\ -2 & 13 & -11 \\ 4 & -11 & 21 \end{bmatrix}$$

On décompose la matrice S puis on vérifie que le résultat obtenu est correct :

```
> L:=choleski(S,3): eval(L),evalm(L&*linalg[transpose](L));
```

$$\begin{bmatrix} 1 & 0 & 0 \\ -2 & 3 & 0 \\ 4 & -1 & 2 \end{bmatrix}, \quad \begin{bmatrix} 1 & -2 & 4 \\ -2 & 13 & -11 \\ 4 & -11 & 21 \end{bmatrix}$$

[Q]

Quelques démonstrations

1. Existence et unicité de la décomposition LU

Proposition [Retour Énoncé]

Une matrice A de $\mathcal{GL}_n(\mathbb{K})$ possède une décomposition LU si et seulement si ses mineurs principaux sont non nuls. Cette décomposition est alors unique.

Démonstration

– *Unicité de la décomposition LU*

Supposons que A ait les décompositions $A = LU$ et $A = L'U'$. Alors $L'^{-1}L = UU'^{-1}$.

Mais les matrices triangulaires supérieures inversibles forment un sous-groupe de $\mathcal{GL}_n(\mathbb{K})$, de même que les matrices triangulaires inférieures à diagonale unité. On en déduit que la matrice $L'^{-1}L = UU'^{-1}$ est à diagonale unité et qu'elle est à la fois triangulaire inférieure et triangulaire supérieure donc diagonale : ce ne peut être que la matrice I_n.

Il en découle $L = L'$ et $U = U'$. On ainsi prouvé l'unicité (si existence.)

– *Existence de la décomposition LU (sens direct)*

On suppose que la matrice inversible A possède une décomposition LU.

On veut montrer que toutes ses sous-matrices principales sont inversibles.

Pour tout k de $\{1, \ldots, n\}$, on utilise une décomposition par blocs de A, L, U :

$$A = \begin{pmatrix} A_k & A'_k \\ A''_k & A'''_k \end{pmatrix}, \quad L = \begin{pmatrix} L_k & 0 \\ L''_k & L'''_k \end{pmatrix}, \quad U = \begin{pmatrix} U_k & U'_k \\ 0 & U'''_k \end{pmatrix}$$

L_k, L'''_k sont triangulaires inférieures à diagonale unité, et U_k, U'''_k sont triangulaires supérieures.

L'égalité $LU = A$ devient $\begin{pmatrix} L_k & 0 \\ L''_k & L'''_k \end{pmatrix} \begin{pmatrix} U_k & U'_k \\ 0 & U'''_k \end{pmatrix} = \begin{pmatrix} A_k & A'_k \\ A''_k & A'''_k \end{pmatrix}$.

Cette égalité implique $L_k U_k = A_k$ par identification au niveau du bloc supérieur gauche.

Or L_k et U_k sont inversibles car leurs coefficients diagonaux, issus de ceux de L et de U, sont non nuls. On en déduit que la matrice A_k est inversible, ce qu'il fallait démontrer.

– *Existence de la décomposition LU (réciproque)*

On suppose que les sous-matrices principales de A sont inversibles, et on veut montrer que A possède une décomposition LU. Pour cela, on raisonne par récurrence sur l'ordre n de A.

Au rang 1 c'est trivial. En effet, si $A = (a)$ avec $a \neq 0$, on a $A = LU$ avec $L = (1)$ et $U = (a)$.

On suppose que la propriété est vraie au rang n, On se donne alors une matrice A inversible d'ordre $n + 1$, et on suppose que toutes les sous-matrices principales de A sont inversibles.

En particulier la sous-matrice A_n d'ordre n est inversible, ainsi que toutes ses sous-matrices principales $A_1, A_2, \ldots, A_{n-1}, A_n$. On en déduit, par hypothèse de récurrence, l'existence (et l'unicité) d'une décomposition $A_n = L_n U_n$.

On cherche alors une décomposition LU de $A_{n+1} = A$ sous la forme $A_{n+1} = L_{n+1}U_{n+1}$, en posant $L_{n+1} = \begin{pmatrix} L_n & 0 \\ R_n & 1 \end{pmatrix}$ et $U_{n+1} = \begin{pmatrix} U_n & C_n \\ 0 & \lambda_n \end{pmatrix}$ où R_n est une matrice-ligne de largeur n, C_n est une matrice-colonne de hauteur n, et λ_n est un scalaire.

Comme dans la démonstration du sens direct, on pose $A = \begin{pmatrix} A_n & A'_n \\ A''_n & A'''_n \end{pmatrix}$.

Dans ce cas particulier, A'_n est une matrice-colonne de hauteur n, A''_n est une matrice-ligne de longueur n, et A'''_n se réduit à un coefficient.

Avec ces notations :

$$L_{n+1}U_{n+1} = A \Leftrightarrow \begin{pmatrix} L_n & 0 \\ R_n & 1 \end{pmatrix}\begin{pmatrix} U_n & C_n \\ 0 & \lambda_n \end{pmatrix} = \begin{pmatrix} A_n & A'_n \\ A''_n & A'''_n \end{pmatrix} \Leftrightarrow \begin{cases} L_nU_n = A_n \\ L_nC_n = A'_n \\ R_nU_n = A''_n \\ R_nC_n + \lambda_n = A'''_n \end{cases}$$

Dans ce système, l'égalité $L_nU_n = A_n$ est déjà connue.

- ◇ L'égalité $L_nC_n = A'_n$ équivaut à $C_n = L_n^{-1}A'_n$.
- ◇ De même, $R_nU_n = A''_n$ équivaut à $R_n = A''_nU_n^{-1}$.
- ◇ La ligne R_n et la colonne C_n étant maintenant connues de manière unique, la dernière égalité du système fournit $\lambda_n = A'''_n - R_nC_n$.

On voit donc qu'il est possible de former la décomposition LU de A, ce qui démontre la propriété au rang $n+1$ et achève la récurrence. [Retour Énoncé]

2. Existence de la décomposition $PA = LU$

Proposition [Retour Énoncé]

Pour toute matrice A de $\mathcal{GL}_n(\mathbb{K})$, il existe au moins une matrice de permutation P telle que la matrice PA admette une décomposition LU.

Démonstration

On procède par récurrence sur l'ordre n de la matrice A.

On note tout d'abord que le résultat est évident si $n = 1$, avec $P = L = (1)$ et $U = A = (a_{11})$.

On suppose donc que la propriété est vraie pour un certain entier $n \geq 1$ et on se donne une matrice carrée A inversible d'ordre $n+1$.

– Montrons tout d'abord qu'il existe une matrice de permutation S telle que la sous-matrice principale B_n d'ordre n de $B = SA$ soit inversible.

  Si la sous-matrice principale A_n d'ordre n de A est inversible, le résultat est évident avec $S = I_{n+1}$. On suppose donc que A_n est non inversible.

  On peut écrire $A = \begin{pmatrix} A_n & C_{n+1} \\ R_{n+1} & a_{n+1,n+1} \end{pmatrix}$, où $\begin{cases} R_{n+1} \text{ est une ligne de largeur } n \\ C_{n+1} \text{ est une colonne de hauteur } n. \end{cases}$

  $T = \begin{pmatrix} A_n \\ R_{n+1} \end{pmatrix}$ est de rang n car formée des n premières colonnes de A inversible.

On peut donc en extraire une sous-matrice carrée inversible d'ordre n. Ce n'est pas A_n, qui est non inversible : la ligne R_{n+1} fait donc partie de cette sous-matrice.

L'échange de R_{n+1} et d'une autre ligne R_k de A_n (avec $1 \leq k \leq n$) permet donc de transformer T en une matrice $T' = \begin{pmatrix} B_n \\ R_k \end{pmatrix}$, où B_n est carrée d'ordre n inversible.

Mais cet échange s'écrit $T' = ST$, où S est la matrice de permutation associée à la bijection de $\{1, 2, \ldots, n+1\}$ qui se contente d'échanger les indices k et $n+1$.

On écrit alors $B = SA = \begin{pmatrix} B_n & D_{n+1} \\ R_k & a_{k,n+1} \end{pmatrix}$ où D_{n+1} est une colonne de hauteur n.

– Avec les notations précédentes, la matrice B_n est inversible. On peut donc lui appliquer l'hypothèse de récurrence : il existe une matrice de permutation Q_n d'ordre n telle que Q_nB_n admette une décomposition L_nU_n, c'est-à-dire telle que toutes les sous-matrices principales de Q_nB_n soient inversibles.

  La matrice Q_n correspond à une permutation de $\{1, 2, ..., n\}$ qu'on étend en une permutation σ de $\{1, 2, ..., n+1\}$ en posant $\sigma(n+1) = n+1$.

  Celle-ci donne alors naissance à la matrice de permutation $Q = Q_\sigma = \begin{pmatrix} Q_n & 0_{n,1} \\ 0_{1,n} & 1 \end{pmatrix}$.

  On a alors $QB = \begin{pmatrix} Q_n & 0_{n,1} \\ 0_{1,n} & 1 \end{pmatrix} \begin{pmatrix} B_n & D_{n+1} \\ R_k & a_{k,n+1} \end{pmatrix} = \begin{pmatrix} Q_nB_n & Q_nD_{n+1} \\ R_k & a_{k,n+1} \end{pmatrix}$.

  La matrice QB est inversible (car Q et B le sont) ainsi que toutes ses sous-matrices principales d'ordre $\leq n$ (car ce sont les sous-matrices principales de Q_nB_n.)

  La matrice QB admet donc une décomposition LU.

– La matrice $P = QS$ est une composée de deux matrices de permutations : c'est donc encore une matrice de permutation. Le calcul précédent montre que PA admet une décomposition LU, ce qui prouve la propriété au rang $n+1$ et achève la récurrence. [Retour Énoncé]

3. Matrices symétriques définies positives

Proposition [Retour Énoncé]

> Soit S une matrice carrée d'ordre n, symétrique et à coefficients réels.
>
> Les conditions suivantes sont équivalentes :
>
> – Toutes les valeurs propres de S sont strictement positives.
>
> – Pour tout vecteur non nul X de $\mathbb{R}^n$, ${}^tXSX > 0$.
>
> – Il existe une matrice inversible M telle que $S = {}^tMM$.

Démonstration

Soit $(\varepsilon) = \varepsilon_1, \varepsilon_2, \ldots, \varepsilon_n$ une base orthonormée de vecteurs propres de S.

Notons λ_k la valeur propre associée à ε_k.

Soit X est un vecteur de $\mathbb{R}^n$, de coordonnées $\alpha_1, \ldots, \alpha_n$ dans la base (ε).

On a $X=\sum\limits_{k=1}^{n}\alpha_k\varepsilon_k$ donc $SX=\sum\limits_{k=1}^{n}\alpha_k\lambda_k\varepsilon_k$ donc $< SX, X >= {}^tXSX=\sum\limits_{k=1}^{n}\lambda_k\alpha_k^2$.

– Si $\lambda_k>0$ pour tout k de $\{1,\ldots,n\}$, alors ${}^tXSX=\sum\limits_{k=1}^{n}\lambda_k\alpha_k^2>0$ pour tout $X\neq\overrightarrow{0}$.

– Réciproquement si ${}^tXSX>0$ pour tout $X\neq\overrightarrow{0}$, alors $X=\varepsilon_k$ donne $\lambda_k={}^tXSX>0$.

– Soit M une matrice carrée d'ordre n, à coefficients réels, inversible.

  La matrice $S=M\,{}^tM$ est symétrique car ${}^tS={}^t(M\,{}^tM))={}^{tt}M\,{}^tM=M\,{}^tM=S$.

  Pour tout vecteur X de $\mathbb{R}^n$, on a ${}^tXSX={}^tXM\,{}^tMX={}^t({}^tMX)({}^tMX)=\|\,{}^tMX\,\|^2$.

  Puisque M (donc ${}^tM$ est inversible) on a $X\neq\overrightarrow{0}\Rightarrow MX\neq\overrightarrow{0}\Rightarrow\|\,MX\,\|>0$.

– Réciproquement, soit S une matrice symétrique P la matrice de passage de la base canonique (e) à une base (ε) orthonormée de vecteurs propres de S.

  Comme les bases (e) et (ε) sont orthonormées, P est une matrice orthogonale : ${}^tP=P^{-1}$.

  On a alors $P^{-1}SP={}^tPSP=D$, où D est diagonale de coefficients diagonaux les λ_k.

  Supposons que les λ_k soient tous strictement positifs.

  Soit Δ la matrice diagonale (inversible) de coefficients diagonaux $\delta_k=\sqrt{\lambda_k}$.

  On a $D=\Delta^2=\Delta\,{}^t\Delta$. Posons alors $M=P\Delta$: cette matrice est inversible.

  On constate que $S=PD\,{}^tP=P\Delta\,{}^t\Delta\,{}^tP=M\,{}^tM$. [Retour Énoncé]

4. Décomposition de Choleski

Proposition [Retour Énoncé]

Soit S une matrice symétrique définie positive. Il existe une unique matrice L triangulaire inférieure à coefficients diagonaux strictement positifs, telle que $S=L\,{}^tL$.

Cette écriture est appelée *décomposition de Choleski* de S.

Démonstration

Par récurrence. Si $n=1$, c'est évident : $S=(\lambda)$, avec $\lambda>0$, et la seule solution est $L=(\sqrt{\lambda})$.

On suppose que la propriété est vraie au rang $n\geq 1$.

Soit S une matrice symétrique définie positive d'ordre $n+1$.

Il faut montrer l'existence et l'unicité de L, triangulaire inférieure à coefficients diagonaux strictement positifs, d'ordre $n+1$ telle que : $S=L\,{}^tL$.

On peut écrire $S=\begin{pmatrix}S_n & {}^tR_n\\ R_n & a\end{pmatrix}$ et chercher L sous la forme $L=\begin{pmatrix}L_n & 0\\ T_n & \lambda\end{pmatrix}$.

Dans cette notation :

– S_n est la sous-matrice principale d'ordre n de S.

– L_n est triangulaire inférieure à coefficients diagonaux strictement positifs, d'ordre n.

- R_n et T_n sont deux vecteurs lignes de taille n.
- a et λ sont deux réels strictement positifs.

On doit prouver l'existence et l'unicité de L_n, T_n et λ.

Avec les notations précédentes :

$$L\,{}^{t}L = S \Leftrightarrow \begin{pmatrix} L_n & 0 \\ T_n & \lambda \end{pmatrix}\begin{pmatrix} {}^{t}L_n & {}^{t}T_n \\ 0 & \lambda \end{pmatrix} = \begin{pmatrix} S_n & {}^{t}R_n \\ R_n & a \end{pmatrix} \Leftrightarrow \begin{cases} L_n\,{}^{t}L_n = S_n & (1) \\ T_n\,{}^{t}L_n = R_n & (2) \\ T_n\,{}^{t}T_n + \lambda^2 = a & (3) \end{cases}$$

Puisque S_n est symétrique définie positive, l'hypothèse de récurrence assure que l'égalité (1) admet une solution L_n unique triangulaire inférieure à coefficients diagonaux > 0.

Cette matrice L_n étant inversible, l'égalité (2) donne alors $T_n = R_n({}^{t}L_n)^{-1}$ de façon unique.

Il reste l'égalité (3), qui s'écrit : $\lambda^2 = a - T_n\,{}^{t}T_n$ (T_n est maintenant connu). Pour l'instant on peut considérer que λ est l'une des deux racines carrées *complexes* du scalaire $a - T_n\,{}^{t}T_n$.

Avec un tel λ provisoire, on a effectivement $S = L\,{}^{t}L$. Mais on sait que $\det(S) > 0$.

On a $\det L = (\det L_n)\lambda$ donc : $\det S = \det(L\,{}^{t}L) = (\det L)^2 = (\det L_n)^2\lambda^2$.

Ainsi $\lambda^2 > 0$. Ceci prouve que λ peut être choisi dans $\mathbb{R}^{+*}$, et ce d'une façon unique.

On a prouvé la propriété au rang $n+1$, ce qui achève la récurrence. [Retour Énoncé]